Les valeurs extrêmes: Test des dérivées secondes

(Partie 1 de 3)

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en février 2006. Ce document Maple est le premier d'une série de trois portant sur les valeurs extrêmes d'une fonction réelle de deux variables réelles. Ce premier document a pour objectif d'utiliser Maple pour trouver les extremums locaux d'une fonction f explicite de deux variables par l'application du test des dérivées secondes.

Rappelons que la fonction à analyser doit d'abord satisfaire à la condition d'applicabilité du test soit

– avoir la continuité des dérivées partielles première et seconde sur un disque centré en (a, b) et

– avoir l'égalité $\frac{\partial}{\partial x} f(x,y)\Big|_{(a,b)} = 0$ et $\frac{\partial}{\partial y} f(x,y)\Big|_{(a,b)} = 0$.

En fait, ce document transpose en Maple la démarche mathématique qui a été présentée en classe.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,display,setoptions3d,spacecurve,pointplot3d):
  setoptions3d(labels=[x,y,z],labels=[x,y,z],lightmodel=light4,axes=
  framed,
                 size=[400,400],axesfont=[TIMES,ROMAN,8],labelfont=
  [TIMES,ROMAN,8]):
```

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimamt les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de
  composition étendue
  Typesetting:-Settings(striptrailing=true);
```

$$extended$$

$$false \qquad \textbf{(1.1)}$$

```
> Digits:=60;
  interface(displayprecision=20);
```

$$Digits := 60$$

$$20 \qquad \textbf{(1.2)}$$

Détermination des valeurs critiques

La première étape dans la recherche des extrema relatifs consiste à rechercher les points critiques (points stationnaires) de la fonction. Un point stationnaire d'une fonction f est un point intérieur du domaine de f qui annule à la fois f_x et f_y ou bien pour lequel au moins une des dérivées partielles n'existe pas.

Soit la fonction f définie par $f(x, y) = x^4 - 3x^2 y + 3y - y^3$.

```
> f:=(x,y)->x^4-3*x^2*y+3*y-y^3:
  'f'(x,y)=f(x,y);
```

$$f(x, y) = x^4 - 3x^2 y - y^3 + 3y \tag{2.1}$$

Obtenons les dérivées partielles f_x et f_y à l'aide de l'opérateur de dérivation D.

```
> fx:=D[1](f):
  'fx'(x,y)=fx(x,y);
  fy:=D[2](f):
  'fy'(x,y)=fy(x,y);
```

$$fx(x, y) = 4x^3 - 6yx$$

$$fy(x, y) = -3x^2 - 3y^2 + 3 \tag{2.2}$$

Remarque: En utilisant l'opérateur de dérivation D, f_x et f_y sont des fonctions des variables x et y.

Maintenant, résolvons simultanément, dans le domaine de définition de f, les équations $f_x(x, y) = 0$ et $f_y(x, y) = 0$. Nous allons résoudre ces équations de deux manières: symboliquement et numériquement (autrement dit: de manière exacte et de manière approximative).

Résolution symbolique (résolution exacte)

La macro-commande `solve` est la macro-commande à utiliser pour la résolution symbolique d'une équation (d'inéquation), d'un ensemble d'équations (d'inéquations) ,

```
> Sol_exactes:=solve({fx(x,y)=0,fy(x,y)=0},{x,y});
```

$$Sol\_exactes := \{x=0, y=1\}, \{x=0, y=-1\}, \{x = RootOf(\_Z^2+3), y=-2\}, \left\{x = \frac{RootOf(\_Z^2-3)}{2}, y = \frac{1}{2}\right\} \tag{2.3}$$

Lorsqu'on utilise la macro-commande `solve`, les racines obtenues peuvent ne pas être données de manière explicite. Afin d'obtenir de façon explicite ces racines formulées en termes de `RooOf`, il suffit d'appliquer la macro-commande `allvalues` sur ces racines.

```
> Sol_exactes:=map(allvalues,{Sol_exactes});
```

$$Sol\_exactes := \left\{\{x=0, y=-1\}, \{x=0, y=1\}, \{x=-I\sqrt{3}, y=-2\}, \{x=I\sqrt{3}, y=-2\}, \left\{x = -\frac{\sqrt{3}}{2}, y = \frac{1}{2}\right\}, \left\{x = \frac{\sqrt{3}}{2}, y = \frac{1}{2}\right\}\right\} \tag{2.4}$$

Remarque: Pour avoir les racines explicitement sans utiliser la macro-commande `allvalues`, nous aurions pu très bien assigner à la variable d'environnement `_EnvExplicit` la valeur logique `true`. Cette assignation pouvant très bien se faire dans la section initialisation de ce document. C'est ce que nous ferons à la partie 3 de cette série sur l'« Optimisation d'une fonction de deux variables ».

Afin de ne retenir que les racines réelles, nous utiliserons l'astuce suivante qui consiste à ajouter (une relation d'ordre) dans le sytème à résoudre l'inéquation $x < \infty$.

```
> Sol_exactes:={solve({fx(x,y)=0,fy(x,y)=0,x<infinity},{x,y})};
```

$$Sol\_exactes := \left\{ \{x=0, y=-1\}, \{x=0, y=1\}, \left\{x=-\frac{\sqrt{3}}{2}, y=\frac{1}{2}\right\}, \left\{x=\frac{\sqrt{3}}{2}, y=\frac{1}{2}\right\} \right\} \quad \textbf{(2.5)}$$

Cette astuce aura permis d'obtenir directement les racines de manière explicite.

Le listage de tous les points critiques est fait automatiquement grâce à la procédure maison suivante. Nous pourrions tout aussi bien établir cette liste manuellement.

Initialisons donc cette petite procédure en exécutant tout le bloc.

```
> Valeurs_Critiques:=proc(Sol_symboliques::set)
  global x,y;
  local Ensemble,k,ordre;
  Ensemble:={};
  for k from 1 to nops(Sol_symboliques) do
  assign(Sol_symboliques[k]);
  Ensemble:=Ensemble union {[x,y]};
  x:='x':y:='y':
  od;
  ordre:=(x,y)->evalb(evalf(op(1,x)<=op(1,y))):
  sort(convert(Ensemble,list),ordre)
  end:
```

Par commodité, donnons donc le nom P à la liste des points critiques.

```
> P:=Valeurs_Critiques(Sol_exactes);
```

$$P := \left[\left[-\frac{\sqrt{3}}{2}, \frac{1}{2} \right], [0, -1], [0, 1], \left[\frac{\sqrt{3}}{2}, \frac{1}{2} \right] \right] \quad \textbf{(2.6)}$$

Les quatres points critiques de la fonction f sont $\left(-\frac{\sqrt{3}}{2}, \frac{1}{2}\right)$, $(0,-1)$, $(0, 1)$, $\left(\frac{\sqrt{3}}{2}, \frac{1}{2}\right)$.

Résolution numérique (résolution approximative)

La macro-commande `fsolve` est à utiliser pour trouver une approximation décimale des racines d'une équation. L'utilisation de cette macro-commande est, en général, plus délicate d'application. En fait, l'efficacité de `fsolve` repose beaucoup sur des valeurs d'amorce qui sont pas "trop" éloignées des racines qu'on cherche à approximer.

```
> Sol_numériques := fsolve({fx(x,y)=0,fy(x,y)=0},{x,y});
```

$$Sol\_numériques := \{x=0., y=-1.\} \quad \textbf{(2.7)}$$

En donnant un intervalle d'amorce approprié, la macro-commande `fsolve` s'avérera plus efficace. Par exemple,

```
> Sol_numériques := fsolve({fx(x,y)=0,fy(x,y)=0},{x,y},{x=-2..0,y=0..1}
  );
```

$$Sol\_numériques := \{x=0., y=1.\} \quad \textbf{(2.8)}$$

Les intervalles de valeurs pour x et y servant d'amorce à la macro-commande `fsolve` ne s'obtiennent évidemment pas par hasard. Un graphique de la situation s'avèrera largement utile.

Application du test des dérivées secondes

Vérifions d'abord l'égalité des dérivées mixtes f_{xy} et f_{yx}.

```
> fxy=D[1,2](f)(x,y);
  fyx=D[2,1](f)(x,y);
```

$$fxy = -6x$$

$$fyx = -6x \tag{3.1}$$

Cette vérification aurait pu se faire directement comme suit.

```
> evalb(D[1,2](f)=D[2,1](f));
```

$$true \tag{3.2}$$

Créons ensuite la formule nécessaire au calcul de $\Delta = f_{xx}(x,y)\, f_{yy}(x,y) - f_{xy}^2(x,y)$ où les dérivées successives d'ordre 2 f_{xx} et f_{yy} définies respectivement par

$f_{xx}(x,y) = \frac{\partial^2}{\partial x^2} f(x,y)$ et par $f_{yy}(x,y) = \frac{\partial^2}{\partial y^2} f(x,y)$ ainsi que la dérivée mixte f_{xy} définie par

$$f_{xy}(x,y) = \frac{\partial^2}{\partial y\, \partial x} f(x,y).$$

```
> fxx:=D[1,1](f);
  fyy:=D[2,2](f);
  fxy:=D[1,2](f);
```

$$fxx := (x,y) \mapsto 12 \cdot x^2 - 6 \cdot y$$

$$fyy := (x,y) \mapsto -6 \cdot y$$

$$fxy := (x,y) \mapsto -6 \cdot x \tag{3.3}$$

Par commodité, créons une fonction qui servira à calculer Δ avec chacune des valeurs critiques:

$$\Delta(x,y) = f_{xx}(x,y)\, f_{yy}(x,y) - f_{xy}^2(x,y)$$

```
> Delta:=(x,y)->fxx(x,y)*fyy(x,y)-(fxy^2)(x,y);
```

$$\Delta := (x,y) \mapsto fxx(x,y) \cdot fyy(x,y) - \left(fxy^2\right)(x,y) \tag{3.4}$$

Présentons, sous la forme d'un tableau, les valeurs de Δ et de f_{xx} en chacune des valeurs critiques.

```
> printf(`\n        Valeurs critiques    |       Delta      |       f
  [xx]`);
  printf(`\n     ============================================================
  ======\n`);
  seq(printf("%25a    | %10a      |%10a \n", P[k],Delta(op(1,op(k,P)),
  op(2,op(k,P))),fxx(op(1,op(k,P)),op(2,op(k,P)))),k=1..nops(P));
```

```
      Valeurs critiques    |       Delta      |       f[xx]
    ============================================================
     [-1/2*3^(1/2), 1/2]    |        -45       |         6
                 [0, -1]    |         36       |         6
                  [0, 1]    |         36       |        -6
      [1/2*3^(1/2), 1/2]    |        -45       |         6
```

Reste donc à conclure: le point critique $(0,-1,f(0,-1))$ est un minimum local, le point critique $(0,1,f(0,1))$ est un maximum local , et les deux autres points crtitiques sont des points-selle: $\left(-\frac{\sqrt{3}}{2}, \frac{1}{2}, f\left(-\frac{\sqrt{3}}{2}, \frac{1}{2}\right)\right)$ et $\left(\frac{\sqrt{3}}{2}, \frac{1}{2}, f\left(\frac{\sqrt{3}}{2}, \frac{1}{2}\right)\right)$.

Illustration de la surface et des points critiques

Illustrons maintenant ces quatre points critiques en les superposant au tracé de la surface.

La difficulté à surmonter avec une telle représentation est, évidemment, de préciser des intervalles appropriés pour les variables *x* et *y* de sorte qu'on puisse inclure tous les points critiques dans la superposition pour que cela puisse se voir de façon clair. L'exemple qui est ici s'y prête bien sans trop de difficultés.

Créons d'abord la structure graphique de cette surface puis affichons-la.

```
> Surface:=plot3d([x,y,f(x,y)],x=-1.5..1.5,y=-1.5..1.5):
  display(Surface,grid=[80,80],orientation=[131,60]);
```



Ensuite, obtenons le tracé de tous les points critiques de la surface en créant d'abord la liste de ces points critiques comme points de l'espace.

```
> Points_surface:=seq([op(1,P[k]),op(2,P[k]),f(op(1,P[k]),op(2,P[k]))],
  k=1..nops(P));
```

$$Points\_surface := \left[-\frac{\sqrt{3}}{2}, \frac{1}{2}, \frac{13}{16}\right], [0, -1, -2], [0, 1, 2], \left[\frac{\sqrt{3}}{2}, \frac{1}{2}, \frac{13}{16}\right] \quad \textbf{(4.1)}$$

La macro-commande `pointplot3d` de la bibliothèque `plots` sera ici très utile. Précisons l'option `patchnogrid` pour mieux voir ces points critiques sur la surface.

```
> Points:=pointplot3d([Points_surface],symbol=solidsphere,color=black):
  display({Surface,Points},lightmodel=none,style=patchnogrid,
  orientation=[131,60]);
```



À l'aide de la macro-commande `spacecurve` de la bibliothèque `plots`, traçons des courbes passant par ces points critiques afin de mettre en évidence la nature de ces extrema locaux.

```
> Courbe_max:=spacecurve([x,1,f(x,1)],x=-1.5..1.5,color=navy,thickness=
  3):
  Courbe_min:=spacecurve([x,-1,f(x,-1)],x=-1.5..1.5,color=orange,
  thickness=3):
  Courbe:=spacecurve([0,y,f(0,y)],y=-1.5..1.5,color=magenta,thickness=
  3):
```

```
> display({Surface,Courbe_max,Courbe_min,Courbe,Points},lightmodel=
  none, style=patchnogrid,orientation=[131,60]);
```



Pour mieux observer la nature de ces points critiques, cliquez sur le graphique précédent en maintenant le bouton gauche enfoncé et modifions l'orientation du graphique.

- Au point $(0, 1, 2)$ dans la direction de la courbe de couleur `navy`, le point critique présente un maximum local et de même dans la direction de la courbe de couleur magenta.
- Au point $(0,-1,-2)$ dans la direction de la courbe de couleur orange, le point critique présente un minimum local et de même dans la direction de la courbe de couleur magenta.

```
> display({Courbe_max,Courbe_min,Courbe,Points},orientation=[131,60]);
```



Traçons maintenant les courbes sur cette surface afin de mettre en évidence la nature des points-selle.

```
> Courbe_A:=spacecurve([-sqrt(3)/2,y,f(-sqrt(3)/2,y)],y=-1.5..1.5,
  color=navy,thickness=3):
  Courbe_B:=spacecurve([sqrt(3)/2,y,f(sqrt(3)/2,y)],y=-1.5..1.5,color=
  orange,thickness=3):
  Courbe:=spacecurve([x,0.5,f(x,0.5)],x=-1.5..1.5,color=magenta,
  thickness=3):
```

```
> display({Surface,Points,Courbe_A,Courbe_B,Courbe},lightmodel=none,
  style=patchnogrid,orientation=[130,60]);
```



– Au point $\left(-\frac{\sqrt{3}}{2}, \frac{1}{2}, \frac{13}{16}\right)$ dans la direction de la courbe de couleur navy, le point critique présente un maximum local et un minimum local dans la direction de la courbe de couleur magenta.

– Au point $\left(\frac{\sqrt{3}}{2}, \frac{1}{2}, \frac{13}{16}\right)$ dans la direction de la courbe de couleur orange, le point critique présente un maximum local et un minimum local dans la direction de la courbe de couleur magenta.

Ces deux points sont donc bien des points-selle.

```
> display({Points,Courbe_A,Courbe_B,Courbe},orientation=[125,80]);
```



Impuissance du test des dérivées secondes

Dans le cas où $\Delta(x, y) = 0$ (ou de la non-existence de $\frac{\partial}{\partial x} \mathrm{f}(x, y)$ ou de $\frac{\partial}{\partial y} \mathrm{f}(x, y)$), le test est impuissant: on ne peut rien conclure quant à la nature des points critiques. Dans ce cas, l'approche graphique est d'une très grande utilité pour examiner la fonction au voisinage du ou des points critiques.

Exemple 1

Soit la fonction f définie par $\mathrm{f}(x, y) = x^3 + y^2$.

```
> f:=(x,y)->x^3+y^2;
```

$$f := (x, y) \mapsto x^3 + y^2 \quad \textbf{(5.1.1)}$$

Obtenons les valeurs critiques de cette fonction.

```
> fx:=D[1](f);
  fy:=D[2](f);
```

$$fx := (x, y) \mapsto 3 \cdot x^2$$

$$fy := (x, y) \mapsto 2 \cdot y \quad \textbf{(5.1.2)}$$

```
> Sol_symboliques:=solve({fx(x,y)=0,fy(x,y)=0,x<infinity},{x,y});
```

$$Sol\_symboliques := \{x = 0, y = 0\} \quad \textbf{(5.1.3)}$$

```
> Sol_symboliques:=map(allvalues,{Sol_symboliques});
```

$$Sol\_symboliques := \{\{x = 0, y = 0\}\} \quad \textbf{(5.1.4)}$$

Nous avons obtenu qu'une seule valeur critique.

```
> P:=Valeurs_Critiques(Sol_symboliques);
```

$$P := [[0, 0]] \quad \textbf{(5.1.5)}$$

Calculons maintenant Δ.

```
> Delta:=(x,y)->D[1,1](f)(x,y)*D[2,2](f)(x,y)-(D[1,2](f)(x,y))^2;
```

$$\Delta := (x, y) \mapsto \mathrm{D}_{1,1}(f)(x, y) \cdot \mathrm{D}_{2,2}(f)(x, y) - \mathrm{D}_{1,2}(f)(x, y)^2 \quad \textbf{(5.1.6)}$$

```
> 'Delta'((0,0))=Delta((0,0));
```

$$\Delta(0, 0) = 0 \quad \textbf{(5.1.7)}$$

Ainsi, puisque $\Delta = 0$, on ne peut rien conclure quant à la nature de cette valeur critique: le test n'est pas applicable. Esquissons alors un graphique de cette fonction au voisinage de $(0, 0)$.

```
> Surface:=plot3d([x,y,f(x,y)],x=-2..2,y=-2..2):
  Points:=pointplot3d([0,0,f(0,0)],symbol=circle,color=black):
  display({Surface,Points},lightmodel=light3,glossiness=0.5,style=
  patchnogrid,orientation=[-150,65]);
```



Pour mieux observer la nature du point $(0, 0, f(0, 0))$, cliquez sur le graphique précédent en maintenant le bouton gauche enfoncé et modifions l'orientation du graphique.

On observe donc que ce point n'est ni un minimum local ni un maximum local. Ce point n'est pas non plus un point-selle.

Traçons des courbes sur cette surface afin de mettre en évidence la nature de ce point critique.

```
> Courbe_A:=spacecurve([0,y,f(0,y)],y=-2..2,color=navy,thickness=3):
  Courbe_B:=spacecurve([x,0,f(x,0)],x=-2..2,color=orange,thickness=
  3):
  display({Surface,Points,Courbe_A,Courbe_B},lightmodel=light3,
  glossiness=0.5,style=patchnogrid,orientation=[-150,65]);
```



On constate clairement que ce point critique n'est pas un point-selle. En effet, dans la direction de la courbe de couleur NAVY, le point critique se présente comme un minimum local mais dans la direction de la courbe de couleur `orange`, le point critique se présente pas comme un maximum local mais plutôt comme un point « plateau », c'est-à-dire comme un point d'inflexion où la tangente est horizontale.

```
> display({Points,Courbe_A,Courbe_B},axes=framed,orientation=[-116,
  65]);
```



Exemple 2

Soit la fonction f définie par $f(x,y) = x^2 + y^4$

```
> f:=(x,y)->x^2+y^4;
```

$$f := (x,y) \mapsto x^2 + y^4 \quad \textbf{(5.2.1)}$$

Cette fois, compte tenu de la règle $x^2 + y^4$, il est assez évident que cette somme possède un seul minimum en $(0,0)$ et que ce minimum est le minimum absolu de cette fonction. Pas besoin d'aide de Maple pour s'en convaincre. Tout de même, faisons l'exercice de développer la démarche du test des dérivées secondes.

Obtenons les valeurs critiques de cette fonction.

```
> fx:=D[1](f);
  fy:=D[2](f);
```

$$fx := (x,y) \mapsto 2 \cdot x$$

$$fy := (x,y) \mapsto 4 \cdot y^3 \quad \textbf{(5.2.2)}$$

```
> Sol_symboliques:=solve({fx(x,y)=0,fy(x,y)=0,x<infinity},{x,y});
```

$$Sol\_symboliques := \{x = 0, y = 0\} \quad \textbf{(5.2.3)}$$

```
> Sol_symboliques:=map(allvalues,{Sol_symboliques});
```

$$Sol\_symboliques := \{\{x = 0, y = 0\}\} \quad \textbf{(5.2.4)}$$

Nous avons obtenu qu'une seule valeur critique.

```
> P:=Valeurs_Critiques(Sol_symboliques);
```

$$P := [[0, 0]] \quad \textbf{(5.2.5)}$$

Calculons maintenant Δ.

```
> Delta:=(x,y)->D[1,1](f)(x,y)*D[2,2](f)(x,y)-(D[1,2](f)(x,y))^2;
```

$$\Delta := (x,y) \mapsto D_{1,1}(f)(x,y) \cdot D_{2,2}(f)(x,y) - D_{1,2}(f)(x,y)^2 \quad \textbf{(5.2.6)}$$

```
> 'Delta'((0,0))=Delta((0,0));
```

$$\Delta(0,0) = 0 \quad \textbf{(5.2.7)}$$

Ainsi, puisque $\Delta = 0$, on ne peut rien conclure quant à la nature de cette valeur critique: le test n'est pas applicable. Esquissons alors un graphique de cette fonction.

```
> Surface:=plot3d([x,y,f(x,y)],x=-2..2,y=-2..2):
  Points:=pointplot3d([0,0,f(0,0)],symbol=circle,color=black):
  display({Surface,Points},lightmodel=none,style=patchnogrid,
  orientation=[-150,45]);
```



On observe clairement que ce point est effectivement un minimum local.

Traçons des courbes sur cette surface afin de mettre en évidence la nature de ce point critique.

```
> Courbe_A:=spacecurve([0,y,f(0,y)],y=-2..2,color=navy,thickness=3):
  Courbe_B:=spacecurve([x,0,f(x,0)],x=-2..2,color=orange,thickness=
  3):
  display({Surface,Points,Courbe_A,Courbe_B},lightmodel=none,style=
  patchnogrid,orientation=[-145,120]);
```




```
> display({Points,Courbe_A,Courbe_B},orientation=[-145,120]);
```



On conclu que le point $(0, 0, f(0, 0))$ est un minimum absolu de la fonction f.

Exemple 3

Soit la fonction f définie par $f(x, y) = -x^2 + y^{\frac{2}{3}}$

```
> f:=(x,y)->-(x^2+y^(2/3));
```

$$f := (x, y) \mapsto -x^2 - y^{2/3} \tag{5.3.1}$$

Avec la règle $-x^2 + y^{\frac{2}{3}}$, c'est moins évident que l'exemple précédent. Obtenons les valeurs critiques de cette fonction.

```
> fx:=D[1](f);
  fy:=D[2](f);
```

$$fx := (x, y) \mapsto -2 \cdot x$$

$$fy := (x, y) \mapsto -\frac{2}{3 \cdot y^{1/3}} \tag{5.3.2}$$

$\frac{\partial}{\partial y} f(x, y)$ n'existe pas quand $y = 0$. Puisque le domaine de la fonction f est $\mathbb{R}^2$, c'est "tout l'axe des x" qui sont de valeurs critiques: $\{(x, 0) | x \in \mathbb{R}\}$. La fonction f possède donc une infinité de valeurs critiques.

Puisque la condition nécessaire d'applicabilité du test n'est pas satisfaite, le test des dérivées secondes est impuissant. Traçons donc le graphique de la surface f en mettant en évidence les points de cette surface qui sont les images des valeurs critiques.

Pour obtenir un tracé correct de la fonction, il est nécessaire de redéfinir la puissance $\frac{2}{3}$ avec la macro-commande `surd` (pourquoi ?).

```
> f:=(x,y)->-(x^2+surd(y^2,3));
  Surface:=plot3d([x,y,f(x,y)],x=-1..1,y=-2..2):
  display({Surface},axes=normal,lightmodel=light2,grid=[80,80],
  orientation=[50,60]);
```

$$f := (x, y) \mapsto -x^2 - \sqrt[3]{y^2}$$



Superposons à la surface tous les points critiques.

```
> Points:=spacecurve([x,0,f(x,0)],x=-1..1,thickness=3,color=orange):
  display({Surface,Points},axes=normal,lightmodel=none,style=
  patchnogrid,orientation=[50,60]);
```



Le graphique montre clairement que $f(0, 0) = 0$ est à la fois un maximum local et le maximum absolu de f sur tout son domaine $\mathbb{R}^2$

```
> display(Points,axes=normal,orientation=[50,60],view=[-1..1,-2..2,
  -1..0]);
```



−2
−1
0
−0.5
−1
0.5
1
0
−0.2
1
2
−0.4
−0.6
−0.8
−1